

Ficheros

- ¿Qué es un fichero?
 - Un fichero es una colección ordenada de elementos, normalmente almacenados en la memoria externa (disco) del que pueden leerse elementos y en el que pueden escribirse nuevos elementos.
 - Permiten el almacenamiento de información de forma permanente (continúa existiendo tras la finalización del programa) y masiva (generalmente mucho mayor que la memoria del ordenador).
- Tipos de ficheros:
 - Según el modo de acceso: ficheros secuenciales y ficheros de acceso directo.
 - En los ficheros secuenciales los elementos pueden leerse únicamente en secuencia: para acceder a un elemento deben haberse leído todos los que le preceden. En los ficheros de acceso directo puede accederse, en cualquier momento, a cualquier elemento.
 - Según el contenido: ficheros binarios y ficheros de texto.
 - Los ficheros binarios contienen datos codificados como cadenas de bits. Los ficheros de texto contienen información que se interpreta como un texto con cierta estructura

Ficheros de texto y binarios

- **Características de los ficheros de texto**
 - Los datos se representan mediante caracteres, divididos en líneas de texto por lo que pueden visualizarse en pantalla.
 - Son más lentos que los ficheros binarios.
 - Tienen acceso secuencial (hay que leer y escribir los datos uno a uno).
 - Solo pueden accederse para lectura o escritura pero no para las dos simultáneamente.
- **Características de los ficheros binarios**
 - Los datos se representan mediante en formato binario por lo que no se visualizan en pantalla.
 - Son rápidos, pues no se necesita conversión entre la representación en memoria de los datos y la representación en el fichero.
 - Tienen acceso aleatorio (puede accederse a cualquier lugar del fichero sin leer o escribir el resto).
 - Pueden accederse para lectura o escritura simultáneamente.
 - Son homogéneos: solo pueden contener un tipo de datos.
 - Suelen ocupar menos espacio que los ficheros de texto

Ficheros de Texto (1)

- Creación de un fichero de texto para escritura

1 Declarar la variable de fichero de tipo `text` (texto)

2 Asociar un nombre de fichero externo a la variable de fichero mediante el procedimiento `assign`:

`assign(vble_fichero, nombre_fichero);`

3 Abrir el fichero mediante una llamada al procedimiento `rewrite`:

`rewrite(vble_fichero);`

4 Escribir el fichero con llamadas a los procedimientos `write` o `writeln`:

`write(vble_fichero, parámetros);`

5 Cerrar el fichero con `close(vble_fichero);`

```
program prueba_texto;
var
  mifichero:text;
  yen,dolar:real;
begin
  yen:=0.0072;
  dolar:=0.75;
  assign(mifichero,'prueba.txt');
  rewrite(mifichero);
  writeln(mifichero,yen:8:3,
  dolar:8:3);
  close(mifichero);
end.
```

Ficheros de texto (2)

- Creación de un fichero de texto para lectura

1 Declarar la variable de fichero de tipo `text` (texto)

2 Asociar un nombre de fichero externo a la variable de fichero mediante el procedimiento `assign`:

```
assign(vble_fichero, nombre_fichero);
```

3 Abrir el fichero mediante una llamada al procedimiento `reset`:

```
reset(vble_fichero);
```

4 Escribir el fichero con llamadas a los procedimientos `read` o `readln`:

```
read(vble_fichero, parámetros);
```

5 Cerrar el fichero con `close(vble_fichero);`

```
program prueba_texto;
var
  mifichero:text;
  yen,dolar:real;
begin
  assign(mifichero,'prueba.txt');
  reset(mifichero);
  readln(mifichero,yen,dolar);
  close(mifichero);
end.
```

Procesamiento de ficheros de texto

- **La función eoln**

```
function eoln(vble_de_fichero);
```

- Devuelve `true` si se ha llegado al final de línea y `false` en otro caso
- No provoca la lectura o escritura de ningún carácter.

- **La función eof**

```
function eof(vble_de_fichero);
```

- Devuelve `true` si se ha llegado al final del fichero y `false` en otro caso
- No provoca la lectura o escritura de ningún carácter.

- **El procedimiento append**

```
procedure append(vble_de_fichero);
```

- Abre un fichero para escritura y comienza a escribir al final del fichero respetando los datos que contuviera el fichero.

Procesamiento de ficheros de texto: Ejemplo

- Contar los caracteres de un fichero de texto

```
program contarcaracteres;
var
  nombrefichero:string;
  fichero:text;
function contador(var fichero:text):
integer
var
  car:char;
  n:integer;
begin
  n:=0;
  while not eof(fichero) do
  begin
    while not eoln(fichero) do
    begin
      read(fichero,car);
      n:=n+1;
    end;
    readln(fichero);
  end;
  contador:=n
end;
```

```
begin
  writeln('Introduzca el nombre del fichero');
  readln(nombrefichero);
  assign(fichero,nombrefichero);
  reset(fichero);
  writeln('El numero de caracteres es: ',
  contador(fichero));
  close(fichero);
end.
```